# $Id: PKGBUILD 79084 2012-10-30 09:55:07Z spupykin $
# Maintainer:
# Contributor: Douglas Soares de Andrade <dsa@aur.archlinux-br.org>
# Contributor: Rick Rein <jeebusroxors@gmail.com>

pkgname=psqlodbc
pkgver=09.01.0200
pkgrel=2
pkgdesc="PostgreSQL ODBC driver"
arch=('i686' 'x86_64')
license=('GPL')
url="http://www.postgresql.org"
depends=('unixodbc' 'postgresql-libs')
makedepends=('postgresql')
options=('!libtool')
source=(ftp://ftp2.it.postgresql.org/mirrors/postgres//odbc/versions/src/$pkgname-$pkgver.tar.gz)
md5sums=('0f4ac0e6b0755af2452764db4cfe4e8f')

build() {
  cd "$srcdir/$pkgname-$pkgver"
  sed -i 's|#define DEFAULT_PGSOCKET_DIR.*|#define DEFAULT_PGSOCKET_DIR "/run/postgresql"|' socket.h
  ./configure --prefix=/usr
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir" install
}
